


ZY886D 煤气表液晶模块

一、概述

ZY886D 液晶模块是本公司生产的煤气表液晶模块，显示信息齐全、优美大方，具有接口简单，工作电源可选 3.6V 或 5V，低功耗等优点，是煤气表配套显示部件的首选。

二、LCD 外观及真值表


液晶片：(77.0mm×25.5mm)



真值表：


PIN	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16
SEG	S0	S1	S2	S3	S4	S5	S6	S7	S8	S9	S10	S11				
COM0	K1	K3	1f	1a	2f	2a	3f	3a	4f	4a	5f	5a				COM0
COM1	K2	K5	1g	1b	2g	2b	3g	3b	4g	4b	5g	5b			COM1	
COM2	K4	K6	1e	1c	2e	2c	3e	3c	4e	4c	5e	5c		COM2		
COM3			1d		2d		3d		4d	Dp	5d	m ³	COM3			

模块尺寸：(98mm×33mm)



三、液晶模块与单片机接口

以与 PHILIPS 的 51LPC 单片机接口为例，原理图如下：



如果要使用/IRQ 信号，请在/IRQ 接口线加一个 5~10K 的上接电阻。

MCU 设置为内部上电复位，复位 IO 口为高，内部 6M 振荡器，WDT 关，12 CLOCK，掉电检测为 2.5V。在不对 LCM 进行数据传送时，请把对应接口 IO 置为高阻态或高电平，以降低 LCM 的功耗。

(演示程序见附录)

四、HT1621 功能详细说明

HT1621 的基本特性


- 操作电压：2.4V~5.2V
- 内置 256KHz RC 振荡器，也可使用 32.768KHz 晶振或 256 KHz 外部时钟源
- 掉电 (Power down) 模式可降低功耗
- 内置 32×4 位显示 RAM
- 3 线串行接口
- 读/写地址自动增加

HT1621 的 AC 特性

符号	参数	测试条件		最小	典型	最大	单位
		V _{DD}	条件				
f _{sys1}	系统时钟	3V	片内 RC 振荡器	—	256	—	kHz
		5V		—	256	—	
f _{sys2}	系统时钟	3V	晶振	—	32.768	—	kHz
		5V		—	32.768	—	
f _{sys3}	系统时钟	3V	外部时钟源	—	256	—	kHz
		5V		—	256	—	
f _{LCD}	LCD 时钟	—	片内 RC 振荡器	—	f _{sys1} /1024	—	Hz
		—	晶振	—	f _{sys2} /128	—	Hz
		—	外部时钟源	—	f _{sys3} /1024	—	Hz
t _{COM}	LCD 公共周期	—	n: COM 数	—	n/f _{LCD}	—	s
f _{CLK1}	串行数据时钟 (/WR)	3V	占空比 50%	—	—	150	kHz
		5V		—	—	300	
f _{CLK2}	串行数据时钟 (/RD)	3V	占空比 50%	—	—	75	kHz
		5V		—	—	150	


f_{TONE}	音调频率	—	片内 RC 振荡器	—	2.0 或 4.0	—	kHz
t_{CS}	串行接口复位脉冲宽度	—	/CS	—	250	—	ns
t_{CLK}	/WR, /RD 输入脉冲宽度	3V	写模式	3.34	—	—	us
			读模式	6.67	—	—	
		5V	写模式	1.67	—	—	us
			读模式	3.34	—	—	
t_r, t_f	串行数据时钟上升/下降沿宽度	3V	—	—	120	—	ns
		5V	—	—			
t_{su}	DATA 到/WR, /RD 时钟建立时间	3V	—	—	120	—	ns
		5V	—	—			
t_h	DATA 到/WR, /RD 时钟建立时间	3V	—	—	120	—	ns
		5V	—	—			
t_{su1}	/CS 到/WR, /RD 时钟建立时间	3V	—	—	100	—	ns
		5V	—	—			
t_{h1}	/CS 到/WR, /RD 时钟建立时间	3V	—	—	100	—	ns
		5V	—	—			

AC 特性时序图



显示存储区—RAM

HT1621 的静态显示存储区 (RAM) 为 32×4 位, 用于保存显示数据。RAM 的内容和 LCD 驱动器的内容相同。可使用 READ, WRITE 和 READ-MODIFY-WRITE 命令对 RAM 中的数据进行访问。从 SEG0 到 SEG31 地址按顺序增加。SEG0 的 6 位地址为 000000, SEG31 地址为 011111。下面所示为 RAM 的映射图。



系统振荡器

HT1621 系统时钟用于产生时基/看门狗定时器 (WDT) 时钟频率、LCD 驱动时钟和声音频率。片内 RC 振荡器 (256KHz)、晶振(32.768KHz)或一个外接由软件设定的 256KHz 时钟可以产生时钟源。系统振荡器配置图参见下图。执行 SYS DIS 命令可以停止系统时钟和 LCD 偏压发生器工作, SYS DIS 命令只适用于片内 RC 振荡器或晶振。当系统时钟停止工作, LCD 将显示空白, 时基/看门狗定时器功能也将失效。

LCD OFF 命令用于关闭 LCD 偏压发生器, 当 LCD 偏压发生器关闭后, 可用 SYS DIS 命令降低系统功耗, 这时 SYS DIS 为降低系统功耗命令。如果系统时钟为外部时钟, SYS DIS 命令不能用于关闭系统时钟和降低系统功耗: 在晶振模式下, 可以外接 32KHz 频率信号源于 OSCI 管脚上, 但是系统不可以进入省电模式; 在外部 256KHz 时钟源模式下, 系统也不能进入省电模式。系统开始上电时, HT1ZY886 处于 SYS DIS 状态。为了使 LCM 功耗为 0, 可以通过切断其电源得到。


时基和看门狗定时器 (WDT)

时基发生器是一个 8 阶增值尖峰计数器, 可以产生准确的时基。看门狗定时器由时基发生器和一个 2 阶增值尖峰计数器组成, 它可以在主控制器或其它子系统处于异常状态时产生中断。WDT 溢出时产生片内 WDT 溢出标志。可用一命令选项可以使时基发生器和 WDT 溢出标志输出到 IRQ 管脚。时基发生器和 WDT 时钟频率有 8 种, $f_{WDT}=32KHz/2^n$, 这里的 n 值为 0~7, 由命令项决定; 等式中 32KHz 为系统频率, 由晶振 (32.768KHz)、片内振荡器 (256KHz)或外部时钟 (256KHz) 产生。


如果系统源频率是片内振荡器频率 (256KHz) 或外部 256KHz 时钟频率, 则被 3 阶预分频器分成 32KHz。时基发生器和 WDT 共用同样的 8 阶计数器, 所以使用与时基发生器和 WDT 相关的命令项时一定要小心。

HT1621 操作时序图


写模式 (命令代码: 101)




写模式（连续写）



命令模式（命令代码：100）



命令模式（数据和命令模式）



HT1621 命令汇总表

名称	ID	命令模式	D/C	功能	Def.
READ	110	A5A4A3A2A1A0D0D1D2D3	D	从 RAM 中读出数据	
WRITE	101	A5A4A3A2A1A0D0D1D2D3	D	将数据写入 RAM	
READ-MODIFY-WRITE	101	A5A4A3A2A1A0D0D1D2D3	D	对 RAM 进行读写	
SYS DIS	100	0000-0000-X	C	关闭系统振荡器和 LCD 偏置发生器	Yes
SYS EN	100	0000-0001-X	C	打开系统振荡器	
LCD OFF	100	0000-0010-X	C	关闭 LCD 偏置发生器	Yes

LCD ON	100	0000-0011-X	C	打开 LCD 偏置发生器	
TIMER DIS	100	0000-0100-X	C	禁止时基输出	
WDT DIS	100	0000-0101-X	C	禁止 WDT 溢出标志输出	
TIMER EN	100	0000-0110-X	C	使能时基输出	
WDT EN	100	0000-0111-X	C	使能 WDT 溢出标志输出	
TONE OFF	100	0000-1000-X	C	关闭音频输出	Yes
TONE ON	100	0000-1001-X	C	打开音频输出	
CLR TIMER	100	0000-11XX-X	C	清零时基发生器内容	
CLR WDT	100	0000-111X-X	C	清零 WDT	
XTAL 32K	100	0001-01XX-X	C	系统时钟源：晶振	
RC 256K	100	0001-10XX-X	C	系统时钟源：片内 RC 振荡器	Yes
EXT 256K	100	0001-11XX-X	C	系统时钟源：外部时钟源	
BIAS 1/2	100	0010-abX0-X	C	LCD 1/2 偏置选项 ab=00: 2 背级 ab=01: 3 背级 ab=10: 4 背级	
BIAS 1/3	100	0010-abX1-X	C	LCD 1/3 偏置选项 ab=00: 2 背级 ab=01: 3 背级 ab=10: 4 背级	
TONE 4K	100	010X-XXXX-X	C	音调频率 4KHz	
TONE 2K	100	011X-XXXX-X	C	音调频率 2KHz	
IRQ DIS	100	100X-0XXX-X	C	禁止 IRQ 输出	Yes
IRQ EN	100	100X-1XXX-X	C	使能 IRQ 输出	
F1	100	101X-X000-X	C	时基/WDT 时钟输出：1Hz WDT 溢出标志：每 4s	
F2	100	101X-X001-X	C	时基/WDT 时钟输出：2Hz WDT 溢出标志：每 2s	
F4	100	101X-X010-X	C	时基/WDT 时钟输出：4Hz WDT 溢出标志：每 1s	
F8	100	101X-X011-X	C	时基/WDT 时钟输出：8Hz WDT 溢出标志：每 1/2s	
F16	100	101X-X100-X	C	时基/WDT 时钟输出：16Hz WDT 溢出标志：每 1/4s	
F32	100	101X-X101-X	C	时基/WDT 时钟输出：32Hz WDT 溢出标志：每 1/8s	
F64	100	101X-X110-X	C	时基/WDT 时钟输出：64Hz WDT 溢出标志：每 1/16s	
F128	100	101X-X111-X	C	时基/WDT 时钟输出：128Hz WDT 溢出标志：每 1/32s	Yes
TEST	100	1110-0000-X	C	测试模式，用户不用	
NORMAL	100	1110-0011-X	C	正常模式	Yes

注：X：无关位

A5~A0: RAM 地址

D3~D0: RAM 数据

D/C: 数据/命令模式

Def: 上电复位默认

所有的粗体 110, 101 和 100 都是模式命令。其中, 100 表示命令模式 ID。如果使用连续命令模式, 第一个命令之后的命令模式 ID 都被忽略。音调频率和时基/WDT 的时钟可由片内 256KHz RC 振荡器、32.768KHz 晶振或外部 256KHz 时钟驱动。建议在上电复位后使用主控器对 HT1621 进行初始化。

(上述介绍的/CS、/RD、/WR、DATA 等分别对应于液晶模块的/CS、/RD、/WR、DA 等)

附录: ZY886D 演示程序

;程序清单

```
*****  
;文件名:ZY886D.asm  
;项目: ZY886D 型 LCD 显示模块演示程序(1.0 版)  
;描述: 启动程序即进入正常显示状态 (循环显示一系列数字 0~F), 依此循环。  
;最后更新日期:2002/02/27  
*****
```

;特殊功能寄存器 SFR 和位地址定义

AUXR1	DATA	0A2H	;辅助功能寄存器
DIVM	DATA	95H	;MCU 时钟除数控制寄存器,
P0M1	DATA	84H	;0 口工作模式选择寄存器 1
P0M2	DATA	85H	;0 口工作模式选择寄存器 1
P1M1	DATA	91H	
P1M2	DATA	92H	
PT0AD	DATA	0F6H	;0 口数字输入功能禁止寄存器

;定义显示数据计数变量

DCONT	DATA	23H	;用于显示位的计数
-------	------	-----	-----------

```
*****  
;  
; 程序开始  
*****
```

```
ORG 0000H  
AJMP RESET
```

```
ORG 0080H
```

RESET:

```
MOV DIVM,#1 ;CPU 工作频率降至正常时的 1/4  
MOV SP,#56H ;设堆栈指针  
MOV TMOD,#01H ;使能定时器 0
```

;显示数据计数器初始化

MOV DCONT,#0

MAIN:

;LCD 工作方式设定

ACALL INTI1621

MOV LCDCOMMAND,#Tone4k ;设置音频输出频率为 4KHZ

ACALL Send1621Com ;由 U1 输出音频信号

;进入正常显示状态

;正常显示情况下，不发送数据时，把 SPI 口均置为高

ACALL RUNOK

;调试子程序

;进入全显示状态

;ACALL RUNALL

;进入掉电模式

;ACALL RUNDOWN

SJMP MAIN

;RUNOK

;从 0 到 f 循环显示 LCD 数位(数字前移)

RUNOK:

ACALL LOADDAT ;装载数据

CLR C

MOV A,#9

SUBB A,DCONT ;判断 DCONT 是否大于 9，大于 9 则打开蜂鸣器

JNC RUNOK_L1

MOV LCDCOMMAND,#ToneOn

ACALL Send1621Com

RUNOK_L1:

ACALL BrushLcd

ACALL DELAY_1S

MOV LCDCOMMAND,#ToneOff ;关闭蜂鸣器

ACALL Send1621Com

ACALL DELAY_1S ;延时

ACALL DELAY_1S

ACALL DELAY_1S

INC DCONT ;计数器加 1

CLR C

MOV A,#11H ;若 DCONT 的值大于 11H，则复位为 0

SUBB A,DCONT


```

JNC     RUNOK_L2
MOV     DCONT,#0
ACALL  INTI1621           ;重新设置 1621
MOV     LCDCOMMAND,#Tone4k ;设置音频输出频率为 4KHZ
ACALL  Send1621Com

```

```

RUNOK_L2:
    SJMP  RUNOK

```

;数据装载程序

;将计数数据进行字模转换后放入显示缓冲区，包括状态字段。

LOADDAT:

```

    MOV     R0,#DISBUFFER           ;指定显示地址为 SEG0 开始
    MOV     R1,DCONT
    MOV     R7,#6                   ;写入 30 个数据(30 个 8 字)

```

LOAD_L1: ;将数据填充显示缓冲区

```

    MOV     A,R1
    MOV     DPTR,#LCDDData
    MOVC   A,@A+DPTR               ;字模转换
    MOV     @R0,A
    INC     R0                       ;缓冲区的下一个地址
    INC     R1
    CLR     C
    MOV     A,R1
    SUBB   A,#12H                   ;0-F 为 10H，加上 2 个空字符等于 12H
    JC     LOAD_L2
    MOV     R1,#00H

```

LOAD_L2:

```

    DJNZ   R7,LOAD_L1

```

```

    ORL    DISBUFFER+4,#01H         ;点亮小数点
    ORL    DISBUFFER+5,#01H         ;点立方米标志
    RET

```

;RUNALL

;全显示输出，调试用

RUNALL:

```

    MOV     R7,#32
    MOV     R0,#DISBUFFER

```

RUNALL_L1:

```

    MOV     @R0,#0FFH               ;装入数据 0FFH
    INC     R0

```

```

    DJNZ    R7,RUNALL_L1

    ACALL   BrushLcd           ;刷新 LCD 显示
    SJMP    $

;RUNDOWN
;掉电方式，调试用
RUNDOWN:
    ACALL   HT1621DOWN        ;进入掉电状态
    SJMP    $

;延时 1S 子程序
DELAY_1S:
    SETB    TR0               ;启动定时器
    MOV     R0,#2
BACK:
    MOV     TL0,#0B0H         ;置初值
    MOV     TH0,#3CH
DL:
    JNB     TF0,DL
    CLR     TF0
    DJNZ    R0,BACK          ;等待定时时间到
    RET

;包含 1621 操作软件包
#include(ZY886D.INC)          ;包含头文件，头文件内已定义 SPI 接口，显示缓冲区，
                              ;及各操作变量，ZY886D 型 LCD 数字显示表等

END

```

库文件原代码:

;LCD 设定（命令模式字的低 8 位数据）

```

BIAS      EQU    52H        ;1/3 偏压 4 背极
RC256     EQU    30H        ;系统时钟选用片内 RC 振荡器
SYSEN     EQU    02H        ;打开系统时钟
LCDON     EQU    06H        ;打开偏压发生器
SYSDIS    EQU    00H        ;关闭系统时钟和偏压发生器
X32K     EQU    28H        ;系统时钟选择为片外晶振
ToneOn    EQU    12H        ;打开音频输出
ToneOff   EQU    10H        ;关闭音频输出

```

Tone4k	EQU	80H	;音频输出频率为 4K
Tone2k	EQU	0C0H	;音频输出频率为 2K
TIMEREN	EQU	0CH	;时基输出使能
TIMERDIS	EQU	08H	;时基输出禁能
CLRtime	EQU	18H	;清零进基发生器
WDTEN	EQU	0EH	;使能 WDT 溢出标志输出
WDTDIS	EQU	0AH	;禁止 WDT 溢出标志输出
CLRWDT	EQU	1CH	;清零 WDT
IRQEN	EQU	10H	;使能 IRQ 输出
IRQDIS	EQU	00H	;禁止 IRQ 输出
FR1	EQU	40H	;设置时基输出 1HZ, WDT 溢出时间 4s

;HT1621B 控制位 (液晶模块接口定义, 用户根据自己的需要更改)

CS	BIT	P0.1	;HT1621B 片选
WRITE	BIT	P0.3	;HT1621B 写控制位
SDATA	BIT	P0.5	;HT1621B 数据线

;显示缓冲区 (用户根据自己的需要更改变量的存储地址)

DISBUFFER	EQU	34H	;显示缓冲区地址 34H--43H(16 字节)
-----------	-----	-----	--------------------------

;临时变量定义(不会破坏这些资源, 只是临时占用, 用户不要自行修改)

lcdbitcnt	EQU	03H	;发送位计数器
lcdcommand	EQU	02H	;发送命令缓冲器
lcdadr	EQU	01H	;lcd 显存指针
LCDBYTECNT	EQU	04H	;发送字节数计数器

=====

;向 HT1621B 发送命令和数据模块

;软件包面向用户的界面程序:

; Send1621Com 发送命令子程序, 命令 ID 为 100+0 型式 + 命令字

; Send1621ComB 发送命令子程序 (B), 命令 ID 为 100+1 型式 + 命令字

; HT1621DOWN 设置 1621 进行掉电工作方式

; INT1621 初化 1621, 并打开 LCD 显示, 用户可更改相应初化字, 进行 1621 的不同设置。

; BrushLcd 刷新 LCD 显示, 即将显示缓冲区的 16 字节数据发送到 1621, 从 SEG0 段开始。

; DISDPTR 将 DPTR 指向的程序存储器的数据 (连续 16 字节) 发送到 1621 显示。

;这此子程序都会自动保护所需的临时单元, 所以占用较多的堆栈, 用户要注意这一点。

=====

;发送命令程序

;名称: Send1621Com

;描述: 向 HT1621B 发送控制命令, 命令内容在 LCDCOMMAND 中。

;所发的命令类型为 ID=100+0+命令, 占用堆栈。

Send1621Com:

PUSH ACC

```

PUSH    LCDBITCNT
ACALL   INTI                ;起动 1621(即初始化)
MOV     LCDBITCNT,#04h
MOV     A,#80h              ;发送 ID 值 100+0
ACALL   sendb0
MOV     A,LCDCOMMAND       ;取得命令字
ACALL   send1621byte       ;发送到 1621
SETB    write
SETB    cs
SETB    sdata
POP     LCDBITCNT
POP     ACC
RET

```

;发送命令程序 (B)

;名称: Send1621ComB

;描述: 向 HT1621B 发送控制命令,命令内容在 LCDCOMMAND 中。

;所发的命令类型为 ID=100+1+命令, 占用堆栈。

Send1621ComB:

```

PUSH    ACC
PUSH    LCDBITCNT
ACALL   INTI                ;起动 1621(即初始化)
MOV     LCDBITCNT,#04h
MOV     A,#90h              ;发送 ID 值 100+0
ACALL   sendb0
MOV     A,LCDCOMMAND
ACALL   send1621byte
SETB    write                ;将 SPI 接口均置为高电平
SETB    cs
SETB    sdata
POP     LCDBITCNT
POP     ACC
RET

```

;HT1621B 掉电模式

;使 1621 进行掉电模式, 以降低功耗

HT1621DOWN:

```

PUSH    LCDCOMMAND
MOV     LCDCOMMAND,#SYSDIS ;关闭系统振荡器即进行掉电方式
ACALL   Send1621Com
POP     LCDCOMMAND
RET

```

;HT1621 设置

;描述: 进行 1621 的基本设置, 偏压、背极、系统工作方式

;用户添加初始化内容或更改设置内容。

INTI1621:

```
PUSH    LCDCOMMAND
MOV     LCDCommand,#bias
ACALL   Send1621Com      ;工作于 1/3 偏压, 4 背极
MOV     LCDCommand,#X32K
ACALL   Send1621Com      ;系统时钟选择为外部时钟
MOV     LCDCommand,#wdtdis
ACALL   Send1621Com      ;关闭看门狗
MOV     LCDCommand,#timerdis
ACALL   Send1621Com
MOV     LCDCommand,#sysen  ;启动 1621 系统振荡器
ACALL   Send1621Com
MOV     LCDCommand,#LCDON  ;启动 1621 显示偏压发生器,启动显示
ACALL   Send1621Com
POP     LCDCOMMAND
RET
```

;刷新显示子程序

;名称: BrushLcd

;描述: 把显示缓冲区 DISBUFFER 共 16 个单元的内容送到 1621 中

BrushLcd:

```
PUSH    LCDBYTECNT
PUSH    LCDADR
MOV     LCDBYTECNT,#16    ;发送 16 个数据
MOV     LCDADR,#DISBUFFER ;指向显示缓冲区
```

DISPLAY0:

```
LCALL   SEND1621DATA      ;调用发送数据子程序, 从 SEG0 段开始
POP     LCDADR
POP     LCDBYTECNT
RET
```

;函数名:DISDPTR

;功能:把 DPTR 指向的程序存储器的 16 个字节内容直接送到 1621 的显示缓冲区内

;参数: DPTR-缓冲区指针

;返回值:无

DISDPTR:

```
PUSH    07H
PUSH    00H
PUSH    06H
PUSH    ACC
MOV     R7,#10H          ;设置各指针及变量初值
MOV     R0,#DISBUFFER   ;先把数据读入显示缓冲区
```

```

MOV    R6,#00
ICERRORCON:
MOV    A,R6
MOVC   A,@A+DPTR      ;读取程序区的数据
MOV    @R0,A          ;放入显示缓冲区
INC    R0
INC    R6
DJNZ   R7,ICERRORCON
ACALL  BrushLcd       ;输出显示
POP    ACC
POP    06H
POP    00H
POP    07H
RET

;*****
;                               子程序
;用户尽量不要单独的调用这此子程序
;*****

;*****
;发送数据程序
;名称: Send1621Data
;描述: 向 HT1621B 发送 BYTECNT 个数据(从 SEG0 开始发送)
;      数据在发送缓冲区中,缓冲区指针 lcdadr.
;-----
Send1621Data:
PUSH   ACC
PUSH   LCDBITCNT
PUSH   LCDBYTECNT
PUSH   LCDADR
ACALL  INTI
MOV    LCDBITCNT,#01H  ;发送写 1621RAM 命令
MOV    A,#80H          ;发送 1 位数据
ACALL  SENDB0
MOV    A,#40H          ;1621RAM 地址设置为 00H
ACALL  SEND1621BYTE
MOV    R1,LCDADR
SENDNEXT1:
MOV    A,@R1          ;读取数据
ACALL  SEND1621BYTE   ;发送数据
INC    R1
DJNZ   LCDBYTECNT,SENDNEXT1

```

```
SETB    WRITE
SETB    CS
SETB    SDATA
POP     LCDADR
POP     LCDBYTECNT
POP     LCDBITCNT
POP     ACC
RET
```

;1621 初始化子程序

;即起动 1621

INTI:

```
CLR     CS
SETB    WRITE
NOP
CLR     WRITE
NOP
RET
```

;发送字节数据子程序

;名称: Send1621Byte

;描述: 向 HT1621B 发送一个字节数据,数据在 A 中

Send1621Byte:

```
PUSH    LCDBITCNT
MOV     lcdbitcnt,#08h      ;发送 8 位数据位
ACALL   SENDB0             ;调用发送数据位子程序
POP     LCDBITCNT
RET
```

;发送位数据子程序

;名称: sendb0

;描述: 向 HT1621B 发送 N 位数据,数据在 A 中,N 在 LCDBITCNT 中,从高位开始发送

SENB0:

```
PUSH    ACC
PUSH    PSW
PUSH    LCDBITCNT
```

SENB01:

```

RLC A
MOV  SDATA,C          ;先发送高位数据
SETB WRITE           ;产生 WR 信号
NOP
NOP
CLR  WRITE
NOP
NOP
DJNZ LCDBITCNT,SENB01 ;未发送完，继续
POP  LCDBITCNT
POP  PSW
POP  ACC
RET

```

```

;=====

```

```

;表格区

```

```

;ZY886D 型 LCD 数字显示表 (0—F)，供用户转换使用。(以上程序不用此表)

```

```

LCDData:

```

```

DB

```

```

0BEH,06H,7CH,5EH,0C6H,0DAH,0FAH,0EH,0FEH,0DEH,0EEH,0F2H,0B8H,76H,0F8H,0E8H,00H,00H,00H

```

```

;=====

```

```

;

```